/*
 * Copyright (C)2005-2017 Haxe Foundation
 *
 * Permission is hereby granted, free of charge, to any person obtaining a
 * copy of this software and associated documentation files (the "Software"),
 * to deal in the Software without restriction, including without limitation
 * the rights to use, copy, modify, merge, publish, distribute, sublicense,
 * and/or sell copies of the Software, and to permit persons to whom the
 * Software is furnished to do so, subject to the following conditions:
 *
 * The above copyright notice and this permission notice shall be included in
 * all copies or substantial portions of the Software.
 *
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
 * IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
 * AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 * LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING
 * FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER
 * DEALINGS IN THE SOFTWARE.
 */
 package sys.db;

class Sqlite
{
	static var type:Class<cs.system.data.IDbConnection>;
	/**
		Opens a new SQLite connection on the specified path.
		Note that you will need a SQLite ADO.NET Provider (see http://www.mono-project.com/docs/database-access/providers/sqlite/).
		Also note that this will try to open an assembly named `Mono.Data.Sqlite` if it wasn't loaded yet.
	**/
	public static function open(file:String):sys.db.Connection
	{
		var cnxString = 'Data Source=$file';
		if (type == null)
		{
			var t = null;
			var assemblies = cs.system.AppDomain.CurrentDomain.GetAssemblies();
			for (i in 0...assemblies.Length)
			{
				var a = assemblies[i];
				t = a.GetType('Mono.Data.Sqlite.SqliteConnection');
				if (t == null)
					t = a.GetType('System.Data.SQLite.SQLiteConnection');
				if (t != null)
				{
					break;
				}
			}

			if (t == null)
			{
				var asm = cs.system.reflection.Assembly.Load('Mono.Data.Sqlite');
				t = asm.GetType('Mono.Data.Sqlite.SqliteConnection');
			}

			if (t != null)
				type = cast cs.Lib.fromNativeType(t);
		}

		if (type == null)
		{
			throw "No ADO.NET SQLite provider was found!";
		}
		var ret = Type.createInstance(type,[cnxString]);
		ret.Open();
		return cs.db.AdoNet.create(ret,'SQLite');
	}
}
